

Re: Combo box

Source:

<http://www.tech-archive.net/Archive/Word/microsoft.public.word.vba.general/2006-05/msg00740.html>

- *From:* Ron Maultsby <RonMaultsby@xxxxxxxxxxxxxxxxxxxxxxxxxxxxxxxxxxxx>
 - *Date:* Wed, 24 May 2006 12:40:02 -0700
-

Doug

Thank you. I will need to look at this. I know Word but I don't know anything about adding lists and drop down boxes to a Word document.

"Doug Robbins – Word MVP" wrote:

From where is the text coming?

You can use the .AddItem method

Or, this routine loads a listbox with client details stored in a table in a separate document (which makes it easy to maintain with additions, deletions etc.), that document being saved as Clients.Doc for the following code.

On the UserForm, have a list box (ListBox1) and a Command Button (CommandButton1) and use the following code in the UserForm_Initialize() and the CommandButton1_Click() routines

```
Private Sub UserForm_Initialize()  
Dim sourcedoc As Document, i As Integer, j As Integer, myitem As Range,  
m As Long, n As Long  
' Modify the path in the following line so that it matches where you  
saved Clients.doc  
Application.ScreenUpdating = False  
' Open the file containing the client details  
Set sourcedoc = Documents.Open(FileName:="e:\worddocs\Clients.doc")  
' Get the number of clients = number of rows in the table of client  
details less one  
i = sourcedoc.Tables(1).Rows.Count - 1  
' Get the number of columns in the table of client details  
j = sourcedoc.Tables(1).Columns.Count  
' Set the number of columns in the Listbox to match  
' the number of columns in the table of client details  
ListBox1.ColumnCount = j  
' Define an array to be loaded with the client data
```

Re: Combo box

```
Dim MyArray() As Variant
'Load client data into MyArray
ReDim MyArray(i, j)
For n = 0 To j - 1
For m = 0 To i - 1
Set myitem = sourcedoc.Tables(1).Cell(m + 2, n + 1).Range
myitem.End = myitem.End - 1
MyArray(m, n) = myitem.Text
Next m
Next n
' Load data into ListBox1
ListBox1.List() = MyArray
' Close the file containing the client details
sourcedoc.Close SaveChanges:=wdDoNotSaveChanges
End Sub
```

```
Private Sub CommandButton1_Click()
Dim i As Integer, Addressee As String
Addressee = ""
For i = 1 To ListBox1.ColumnCount
ListBox1.BoundColumn = i
Addressee = Addressee & ListBox1.Value & vbCrLf
Next i
ActiveDocument.Bookmarks("Addressee").Range.InsertAfter Addressee
UserForm2.Hide
End Sub
```

The Initialize statement will populate the listbox with the data from the table and then when a client is selected in from the list and the command button is clicked, the information for that client will be inserted into a bookmark in the document. You may want to vary the manner in which it is inserted to suit our exact requirements, but hopefully this will get you started.

To make it easy for you, the code has been written so that it will deal with any number of clients and any number of details about each client. It assumes that the first row of the table containing the client details is a header row.

And, here are a couple of ways of loading data from a table in Access;

```
Private Sub UserForm_Initialize()
Dim db As DAO.Database
Dim rs As DAO.Recordset
Dim NoOfRecords As Long
' Open the database
Set db = OpenDatabase("D:\Access\ResidencesXP.mdb")
' Retrieve the recordset
Set rs = db.OpenRecordset("SELECT * FROM Owners")
' Determine the number of retrieved records
With rs
```

Re: Combo box

```
.MoveLast
NoOfRecords = .RecordCount
.MoveFirst
End With
' Set the number of Columns = number of Fields in recordset
ListBox1.ColumnCount = rs.Fields.Count
' Load the ListBox with the retrieved records
ListBox1.Column = rs.GetRows(NoOfRecords)
' Cleanup
rs.Close
db.Close
Set rs = Nothing
Set db = Nothing
End Sub
```

```
Private Sub UserForm_Initialize()
'allocate memory for the database object as a whole and for the active
record
Dim myDataBase As Database
Dim myActiveRecord As Recordset
'Open a database
Set myDataBase = OpenDatabase("E:\Access97\Ely\ResidencesXP.mdb")
'Access the first record from a particular table
Set myActiveRecord = myDataBase.OpenRecordset("Owners", dbOpenForwardOnly)
'Loop through all the records in the table until the end-of-file marker is
reached
Do While Not myActiveRecord.EOF
ListBox1.AddItem myActiveRecord.Fields("Owner")
'access the next record
myActiveRecord.MoveNext
Loop
'Then close the database
myActiveRecord.Close
myDataBase.Close
End Sub
```

--
Hope this helps.

Please reply to the newsgroup unless you wish to avail yourself of my services on a paid consulting basis.

Doug Robbins – Word MVP

"Leeka" <Leeka@xxxxxxxxxxxxxxxxxxxxxxxxxxxx> wrote in message
news:B35540F2-AED0-4CEC-A4FB-B0D88BB01E03@xxxxxxxxxxxxxxxxxxxx

Need help adding text to combobox and dropdown list box

Re: Combo box